Problem:
Score : 300 points

Problem Statement

We have N switches with "on" and "off" state, and M bulbs. The switches are numbered 1 to N, and the bulbs are numbered 1 to M.

Bulb i is connected to k_i switches: Switch s_{i1}, s_{i2}, ..., and s_{ik_i}. It is lighted when the number of switches that are "on" among these switches is congruent to p_i modulo 2.

How many combinations of "on" and "off" states of the switches light all the bulbs?

Constraints

1 \leq N, M \leq 10

1 \leq k_i \leq N

1 \leq s_{ij} \leq N

s_{ia} \neq s_{ib} (a \neq b)

p_i is 0 or 1.

All values in input are integers.

Input

Input is given from Standard Input in the following format:

N M
k_1 s_{11} s_{12} ... s_{1k_1}
:
k_M s_{M1} s_{M2} ... s_{Mk_M}
p_1 p_2 ... p_M

Output

Print the number of combinations of "on" and "off" states of the switches that light all the bulbs.

Sample Input 1

2 2
2 1 2
1 2
0 1

Sample Output 1

1

Bulb 1 is lighted when there is an even number of switches that are "on" among the following: Switch 1 and 2.

Bulb 2 is lighted when there is an odd number of switches that are "on" among the following: Switch 2.

There are four possible combinations of states of (Switch 1, Switch 2): (on, on), (on, off), (off, on) and (off, off). Among them, only (on, on) lights all the bulbs, so we should print 1.

Sample Input 2

2 3
2 1 2
1 1
1 2
0 0 1

Sample Output 2

0

Bulb 1 is lighted when there is an even number of switches that are "on" among the following: Switch 1 and 2.

Bulb 2 is lighted when there is an even number of switches that are "on" among the following: Switch 1.

Bulb 3 is lighted when there is an odd number of switches that are "on" among the following: Switch 2.

Switch 1 has to be "off" to light Bulb 2 and Switch 2 has to be "on" to light Bulb 3, but then Bulb 1 will not be lighted. Thus, there are no combinations of states of the switches that light all the bulbs, so we should print 0.

Sample Input 3

5 2
3 1 2 5
2 2 3
1 0

Sample Output 3

8

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import Data.List

-- All subsets of [0 .. n-1]
subsets :: Int -> [[Int]]
subsets 0 = [[]]
subsets n = let xs = subsets (n-1) in xs ++ [x ++ [n-1] | x <- xs]

-- Intersection.  Assume that two lists are sorted
intersection :: [Int] -> [Int] -> [Int]
intersection _ [] = []
intersection [] _ = []
intersection xs@(x:xs1) ys@(y:ys1)
  | x == y = x : intersection xs1 ys1
  | x < y  = intersection xs1 ys
  | x > y  = intersection xs ys1

-- Whether a lamp with spec (ks,p) is on when switched in xs are on
check :: (Int, [Int]) -> Int -> [Int] -> Bool
check (k, ss) p xs = length (intersection (sort ss) xs) `mod` 2 == p

-- Whether all lamps are on when switches in xs are on
all_on :: [(Int, [Int])] -> [Int] -> [Int] -> Bool
all_on lst_ks lst_p xs = and [check ks p xs | (ks, p) <- zip lst_ks lst_p]

solve :: Int -> Int -> [(Int, [Int])] -> [Int] -> Int
solve n m lst_ks lst_p
  = length (filter (all_on lst_ks lst_p) (subsets n))

parse_ks :: String -> (Int, [Int])
parse_ks str = let k : lst_s = map read (words str)
               in (k, map (+ (-1)) lst_s)   -- 1-indexed -> 0-indexed

main :: IO()
main = do
  cont <- getContents
  let line1 : line_rest = lines cont
      lines_ks = init line_rest
      line_p = last line_rest
      [s_n, s_m] = words line1
      n = read s_n
      m = read s_m
      lst_ks = map parse_ks lines_ks
      lst_p = map read (words line_p)
      answer = solve n m lst_ks lst_p
  putStrLn (show answer)